Download GAMS Programs:

ftp://ftp.zew.de/pub/zew-docs/div/td-bu.pdf

Part 1: MCP Formulation

```
$Title  Static maquette of integrated TD/BU hybrid model

*       Model formulation in MCP

*============================================================================
* Model code for stylzed integrated bottom-up/top-down analysis of energy
* policies based on:
*
*               ZEW Discussion Paper 05-28
*          Integrating Bottom-Up into Top-Down:
*            A Mixed Complementarity Approach
*
* Contact the authors at: boehringer@zew.de; tom@mpsge.org
*=============================================================================

* For plotting the results you must have installed the gnuplot-shareware
* (see http://debreu.colorado.edu/gnuplot/gnuplot.htm for downloads)

*=============================================================================
*       List of parameters subject to sensitivity analysis
*       The user can change the default settings.

* Choice of key elasticities:
*       Elasticity of substitution in final consumption
$if not setglobal esub_c $setglobal esub_c 0.5

*       Elasticity in gas supply
$if not setglobal esub_gas  $setglobal esub_gas  1.5

*       Elasticity in coal supply
$if not setglobal esub_coal  $setglobal esub_coal 3

*       Elasticity in oil supply
$if not setglobal esub_oil  $setglobal esub_oil 1.5


* Choice of resource availability for renewables:
* (as a fraction of base-year total electricity production)
```

```
*       Potential wind supply - (%)
$if not setglobal p_wind  $setglobal p_wind  10

*       Potential solar supply - (%)
$if not setglobal p_sun  $setglobal p_sun   10

*       Potential biomass supply - (%)
$if not setglobal p_trees  $setglobal p_trees 10


* Cost disadvantage of inital slack technologies:
*       Wind energy premium (%)
$if not setglobal c_wind  $setglobal c_wind 10

*       Solar energy premium (%)
$if not setglobal c_solar  $setglobal c_solar   10

*       Biomass energy premium (%)
$if not setglobal c_biomass  $setglobal c_biomass 10


* Other central model assumptions:
*       Time horizon (short, long)
*       N.B.: For short-run analysis capital is immobile across sectors
$if not setglobal horizon  $setglobal horizon long

*============================================================================


*       Assign user-specific changes of default assumptions
scalar  shortrun        Flag for short-run capital mobility/1/;

$if "%horizon%"=="long" shortrun=0;

*       Elasticititities of substitution (ESUB)
scalar  esub_c     Elasticity of substituion in final demand /%esub_c%/
        esub_ele   ESUB between electricity and oil in final demand /0.5/
        esub_k_e   ESUB between capital and energy in ROI production /0.5/
        esub_l_ke  ESUB between labor and other inputs in ROI production /0.8/;

set     t       Electricity Technologies (current and future)
                /coal,gas,nuclear,hydro,wind,solar,biomass/;

```

```
set     xt(t)   Existing technologies /coal,gas,nuclear,hydro/;

set     nt(t)   New vintage technologies /wind,solar,biomass/;

set     ff      Fossil fuel inputs /coa, gas, oil/;

set     n       Natural resources /wind, sun, trees/;


set     res(t)  Renewable energy sources /hydro, wind, solar, biomass/;

*       The following data table describes an economic equilibrium in
*       the base year:


table sam  Base year social accounting matrix

        roi     coa     gas     oil     ele     ra
roi     200     -5      -5      -10     -10    -170
coa             15                      -15
gas                     15              -15
oil                             30              -30
ele     -10                              60     -50
capital -80                             -20     100
labor   -110    -5      -5      -10             130
rent            -5      -5      -10              20             ;

parameter carbon(ff)    Carbon coefficients /oil 1, gas 1, coa 2/;

scalar  carblim         Carbon target /0/;

parameter  esub_ff(ff)  Elastictity of substitution in fossil fuel production
        /gas %esub_gas%, coa %esub_coal%, oil %esub_oil%/;

*       The following data tables describes electricy generation in
*       the base year as well as the technology coefficients for technologies
*       which are inactive in the base year (wind, solar, biomass). Inactive
*       technologies are by defaults %c_****% more costly.

table xtelec  Electricty technologies - extant (initially active)

        coal    gas     nuclear hydro
ele       20     20          12     8
```

```
roi       -1     -1           -8
gas             -15
coa      -15
capital   -4     -4           -4    -8;


table ntelec  Electricty technologies - new vintage (initially inactive)

         wind    solar   biomass
ele       1.0      1.0      1.0
roi       -.2      -.3      -.4
capital   -.9      -.8      -.7
wind     -1.0
sun               -1.0
trees                      -1.0;


*       Adjust the cost coefficients for initially inactive technologies
*       according to user assumptions:
set xk /roi, capital/;

ntelec(xk,"wind")    = ntelec(xk,"wind")    * (100+%c_wind%)/110;
ntelec(xk,"solar")   = ntelec(xk,"solar")   * (100+%c_solar%)/110;
ntelec(xk,"biomass") = ntelec(xk,"biomass") * (100+%c_biomass%)/110;


*       Specify limits (resource or policy constraints) to the availability
*       of technologies

parameter limit    Electricty supply limits on extant technologies /
        nuclear         12
        hydro            8 /;

parameter  nrsupply(n)  Natural resource supplies (fraction of base output)/
        wind            %p_wind%
        sun             %p_sun%
        trees           %p_trees% /;

nrsupply(n) = nrsupply(n)/100 * sam("ele","ele");

parameter c0    Baseyear final consumption;
c0 = (-sam("roi","ra")-sam("ele","ra")-sam("oil","ra"));

```

```
set     quota(t)  Flag for technologies contributing to green quota;
quota(t) = no;

scalar  share   Target share for green quota /0/;

*       By default we might set target share for green quota at base year level
share = sum(t$res(t), xtelec("ele",t))/sum(t, xtelec("ele",t));
display share;

scalar
        dd      Flag for double dividend policy analysis /0/,
        ls      Flag for lump-sum revenue-recyling       /0/,
        vat     Flag for VAT revenue recycling           /0/,
        g0      Base year public consumption             /0/,
        tc0     Base year consumption tax                /0/;


positive variables
*       Activitiy levels
roi     Aggregate output
ele(t)  Production levels for electricity by technology
s(ff)   Fossil fuel supplies
c       Aggregate consumption (utility) formation
g       Public good provision

*       Price levels
proi    Price of aggregate output
pele    Price of electricty
pf(ff)  Price of oil and gas
pl      Wage rate
pk      Price of malleable capital for X (and NT elec)
pr(ff)  Rent on fossil fuel resources
pn(n)   Rent on natural resources
pc      Consumption (utility) price index
pg      Price of public consumption
plim(t) Shadow price on electricity expansion
pkx(t)  Price of capital to extant technologies
pcarb   Carbon tax rate

*       Income variables
ra      Representative household
govt    Government
```

```
*       Endogenous taxes or subsidies
tau     Uniform subsidy rate on renewable energy;

positive variables
phi_ls  Lump-sum recycling
phi_tc  Consumption tax recycling;


equations

*       Zero profit conditions for activities linked to activity levels
zprf_roi        Zero profit condition for macro production sector
zprf_ele(t)     Zero profit condition for alternative electricity supply technologies
zprf_s(ff)      Zero profit condition for fossil fuel supplies
zprf_c          Zero profit condition for aggregate utility formation
zprf_g          Zero profit condition for public good formation

*       Market clearance conditions for goods linked to prices
mkt_proi        Market clearance condition for macro production good
mkt_pele        Market clearance condition for electricity
mkt_pf(ff)      Market clearance condition for fossil fuels coal and gas
mkt_pl          Market clearance condition for labor
mkt_pk          Market clearance condition for malleable capital
mkt_pr(ff)      Market clearance conditions for fossil fuel resources
mkt_pn(n)       Market clearance conditions for natural resources
mkt_pcarb       Market clearance condition for carbon
mkt_pkx(t)      Market clearance condition for capital inputs to extant power production
mkt_plim(t)     Market clearance condition for capacity on electricity expansion
mkt_pc          Market clearance for aggregate utility good
mkt_g           Market clearance for public good

*       Income balance for representative household linked to income level
inc_ra          Budget constraint for representative household
inc_govt        Budget constraint for government

*       Additional constraints
sub_res         Endogenous subsidy to achieve renewable energy quota
eqy_ls          Equal yield constraint for lump-sum recycling
eqy_tc          Equal yield constraint for consumption tax recycling

parameter
theta_l_roi      Cost share of labor in ROI production
```

```
theta_ele_roi    Cost share of electricity in capital-electricity composite of ROI
theta_r_ff(ff)   Cost share of fossil fuel resource in fossil fuel production
theta_l_ff(ff)   Cost share of labor in non-resource input of fossil fuel production
theta_roi_ff(ff) Cost share of ROI in ROI-labor composite of fossil fuel production
theta_ele_c      Cost share of electricity in oil-electricity composite of final consmption
theta_roi_c      Cost share of ROI in final consumption
theta_roi_t(t)   Cost share of ROI in electricity production by technology t
theta_k_t(t)     Cost share of capital in electricity production by technology t
theta_ff_t(ff,t) Cost share of fossil fuel in electricity production by technology t;

theta_roi_c     = -sam("roi","ra")/c0;
theta_l_roi     = (-sam("labor","roi"))/sam("roi","roi");
theta_ele_roi   = (-sam("ele","roi"))/ ((-sam("capital","roi")) + (-sam("ele","roi")));
theta_r_ff(ff)  = (-sam("rent",ff))/ ((-sam("rent",ff)) + (-sam("roi",ff)) + (-sam("labor",ff)));
theta_roi_ff(ff)= (-sam("roi",ff)) / ((-sam("roi",ff)) + (-sam("labor",ff)));
theta_ele_c     = (-sam("ele","ra"))/((-sam("ele","ra")) + (-sam("oil","ra")));
theta_roi_t(t)$xt(t)    = (-xtelec("roi",t)/xtelec("ele",t));
theta_k_t(t)$xt(t)      = (-xtelec("capital",t)/xtelec("ele",t));
theta_ff_t(ff,t)$xt(t)  = (-xtelec(ff,t)/xtelec("ele",t));
theta_roi_t(t)$nt(t)    = (-ntelec("roi",t)/ntelec("ele",t));
theta_k_t(t)$nt(t)      = (-ntelec("capital",t)/ntelec("ele",t));
theta_l_ff(ff)          = (-sam("labor",ff))/((-sam("labor",ff))+(-sam("roi",ff)));

*       Definition of zero profit conditions
zprf_roi..
    (theta_l_roi*pl**(1-esub_l_ke) + (1- theta_l_roi)
    *(theta_ele_roi*pele**(1-esub_k_e) + (1-theta_ele_roi)*pk**(1-esub_k_e))
    **((1-esub_l_ke)/(1-esub_k_e)))**(1/(1-esub_l_ke))
                =G= proi;

zprf_ele(t)..
    {theta_roi_t(t)*proi+ sum(ff,theta_ff_t(ff,t)*pf(ff))
     + (theta_k_t(t)*pkx(t))$shortrun
     + (theta_k_t(t)*pk)$(not shortrun)
     +  plim(t)$limit(t)
    }$xt(t)
    +
    {theta_roi_t(t)*proi + theta_k_t(t)*pk + sum(n, (-ntelec(n,t))*pn(n))}$nt(t)
             =G= pele*(1+tau$quota(t));

zprf_s(ff)..
    (theta_r_ff(ff)*pr(ff)**(1-esub_ff(ff)) +  (1-theta_r_ff(ff))*( theta_l_ff(ff)*pl
     + (1-theta_l_ff(ff))*proi)**(1-esub_ff(ff)))**(1/(1-esub_ff(ff)))
```

```
    + ((carbon(ff)*pcarb))$carblim
            =G= pf(ff);

zprf_c..
   (theta_roi_c*((proi*(1+tc0*phi_tc$dd))/(1+tc0$dd))**(1-esub_c)
    + (1-theta_roi_c)*(theta_ele_c*pele**(1-esub_ele)
    +(1-theta_ele_c)*pf("oil")**(1-esub_ele))**((1-esub_c)/(1-esub_ele)))**(1/(1-esub_c))
          =G= pc;

zprf_g$dd..
   proi =G= pg;

*       Definition of market clearance conditions
mkt_proi..
       roi*sam("roi","roi")  =G=
       sum(xt, ele(xt)*(-xtelec("roi",xt)/xtelec("ele",xt)))
       + sum(nt, ele(nt)*(-ntelec("roi",nt)))
       + sum(ff, (-sam("roi",ff))*s(ff)* ((theta_r_ff(ff)*pr(ff)**(1-esub_ff(ff))
       +  (1-theta_r_ff(ff))*( theta_l_ff(ff)*pl
       + (1-theta_l_ff(ff))*proi)**(1-esub_ff(ff)))**(1/(1-esub_ff(ff)))
       /( theta_l_ff(ff)*pl  + (1-theta_l_ff(ff))*proi))**esub_ff(ff))
       + (-sam("roi","ra")/(1+tc0$dd))*c*( (pc/(proi*(1+(tc0*phi_tc)$dd)))*(1+tc0$dd))**esub_c
       + (g0*g)$dd;

mkt_pele..
       sum(t, ele(t))  =G=
      (-sam("ele","ra"))*c*(pc/(theta_ele_c*pele**(1-esub_ele)
      +(1-theta_ele_c)*pf("oil")**(1-esub_ele))**(1/(1-esub_ele)))**esub_c
      * (((theta_ele_c*pele**(1-esub_ele)
      +(1-theta_ele_c)*pf("oil")**(1-esub_ele))**(1/(1-esub_ele)))/pele)**esub_ele
      + (-sam("ele","roi"))*roi*(proi/((theta_ele_roi*pele**(1-esub_k_e)
      + (1-theta_ele_roi)*pk**(1-esub_k_e))**(1/(1-esub_k_e))))**esub_l_ke
      *((theta_ele_roi*pele**(1-esub_k_e)
      + (1-theta_ele_roi)*pk**(1-esub_k_e))**(1/(1-esub_k_e))/pele)**esub_k_e;

mkt_pf(ff)..
      sam(ff,ff)*s(ff) =G=
      sum(xt, (-xtelec(ff,xt)/xtelec("ele",xt))*ele(xt))
      + (-sam(ff,"ra"))*c*(pc/(theta_ele_c*pele**(1-esub_ele)
      +(1-theta_ele_c)*pf("oil")**(1-esub_ele))**(1/(1-esub_ele)))**esub_c
      * (((theta_ele_c*pele**(1-esub_ele)
      +(1-theta_ele_c)*pf("oil")**(1-esub_ele))**(1/(1-esub_ele)))/pf("oil"))**esub_ele;
```

```
mkt_pl..
       sam("labor","ra") =G=
       (-sam("labor","roi"))*roi*(proi/pl)**esub_l_ke
       + sum(ff, (-sam("labor",ff))*s(ff)* ((theta_r_ff(ff)*pr(ff)**(1-esub_ff(ff))
       +  (1-theta_r_ff(ff))*( theta_l_ff(ff)*pl
       + (1-theta_l_ff(ff))*proi)**(1-esub_ff(ff)))**(1/(1-esub_ff(ff)))
       /( theta_l_ff(ff)*pl  + (1-theta_l_ff(ff))*proi))**esub_ff(ff));

mkt_pk..
      (-sam("capital","roi")+sum(xt,(-xtelec("capital",xt)))$(not shortrun)) =G=
      (-sam("capital","roi"))*roi*(proi/((theta_ele_roi*pele**(1-esub_k_e)
      + (1-theta_ele_roi)*pk**(1-esub_k_e))**(1/(1-esub_k_e))))**esub_l_ke
      *((theta_ele_roi*pele**(1-esub_k_e)
      + (1-theta_ele_roi)*pk**(1-esub_k_e))**(1/(1-esub_k_e))/pk)**esub_k_e
      + sum(xt$(not shortrun),(-xtelec("capital",xt)/xtelec("ele",xt))*ele(xt))
      + sum(nt,(-ntelec("capital",nt))*ele(nt));

mkt_pr(ff)..
      (-sam("rent",ff)) =G=
      (-sam("rent",ff))*s(ff)* ((theta_r_ff(ff)*pr(ff)**(1-esub_ff(ff))
      +  (1-theta_r_ff(ff))*( theta_l_ff(ff)*pl
      + (1-theta_l_ff(ff))*proi)**(1-esub_ff(ff)))**(1/(1-esub_ff(ff)))/pr(ff))** esub_ff(ff);

mkt_pn(n)..
      nrsupply(n) =G= sum(nt,(-ntelec(n,nt))*ele(nt));

mkt_pkx(xt)$shortrun..
      (-xtelec("capital",xt)) =G= (-xtelec("capital",xt)/xtelec("ele",xt))*ele(xt);

mkt_plim(xt)$limit(xt)..
      limit(xt) =G= ele(xt);

mkt_pcarb$carblim..
      carblim =G= sum(ff,(carbon(ff)*sam(ff,ff))*s(ff));

mkt_pc ..
      c0*c =G= ra/pc ;

mkt_g$dd ..
     g0*g =G= govt/pg ;

*       Income definition for representative household
inc_ra..
```

```
    (-sam("capital","roi")+sum(xt,(-xtelec("capital",xt)))$(not shortrun))*pk
    + sum(xt$shortrun, (-xtelec("capital",xt))*pkx(xt))
    + sam("labor","ra")*pl
    + sum(ff,(-sam("rent",ff))*pr(ff))
    + sum(n, nrsupply(n)*pn(n))
    + (carblim*pcarb)$carblim$(not dd)
    + sum(xt$limit(xt), limit(xt)*plim(xt))
    - sum(t$quota(t), pele*ele(t)*tau)
    - (pc*phi_ls)$dd
    =G=  ra;
*       Income definition for government
inc_govt$dd..
    (carblim*pcarb)$carblim + pc*phi_ls
    + ((-sam("roi","ra")/(1+tc0$dd))*c
    *( (pc/(proi*(1+(tc0*phi_tc)$dd)))*(1+tc0$dd))**esub_c)*proi*tc0*phi_tc
    =G= govt;
*       Endogenous subsidy to assure renewables quota
sub_res$card(quota)..
    sum(t$res(t), ele(t)) =G= share*sum(t, ele(t));
*       Endogenous equal yield constraints
eqy_ls$dd..
     g =G= 1;
eqy_tc$dd..
     g =G= 1;
*       Define MCP model
model mcp_hybrid / zprf_roi.roi, zprf_ele.ele, zprf_s.s, zprf_c.c, zprf_g.g,
                mkt_proi.proi, mkt_pele.pele, mkt_pf.pf, mkt_pl.pl,
                mkt_pk.pk, mkt_pr.pr, mkt_pn.pn, mkt_pcarb.pcarb,
                mkt_pkx.pkx, mkt_plim.plim, mkt_pc.pc, mkt_g.pg, inc_ra.ra,
                sub_res.tau, inc_govt.govt, eqy_ls.phi_ls, eqy_tc.phi_tc
                /;
*       Benchmark initialization
*       In the base year new-vintage technologies are inactive
*       and the prices of backstop natural resources are zero
*       Extant technologies with capacity limits are assumed to
```

```
*       operate at the upper bound with a zero shadow value in the
*       base year

ele.l(nt)  = 0;
pn.l(n)    = 0;
plim.l(xt) = 0;

ele.l(xt)  = xtelec("ele",xt);

*       Initialize activities and prices
roi.l = 1; ele.l(xt)= xtelec("ele",xt);  s.l(ff) = 1; c.l = 1;
proi.l = 1; pele.l = 1; pf.l(ff) = 1; pl.l = 1; pk.l = 1; pr.l(ff) = 1;
pkx.l(t)$((-xtelec("capital",t))$shortrun) = 1; plim.l(t) = 0;
pn.l(n) = 1; pc.l = 1;

*       Install lower bounds on prices to avoid divison by zero in MCP formulation
proi.lo = 1e-5; pele.lo = 1e-5; pf.lo(ff) = 1e-5; pl.lo = 1e-5; pk.lo = 1e-5;
pr.lo(ff) = 1e-5; pkx.lo(t)$((-xtelec("capital",t))$shortrun) = 1e-5; pc.lo = 1e-5;

*       Tie down "active" model specification
phi_tc.fx = 1; phi_ls.fx = 0;
g.fx = 0; pg.fx = 0; govt.fx = 0; pcarb.fx = 0;
pkx.fx(t)$(not (-xtelec("capital",t))$shortrun) = 0;
tau.fx$(not card(quota)) = 0;
plim.fx(t)$(not limit(t)) = 0;

*       In the base year we have no new-vintage electricity and the prices of backstop
*       natural resources are zero:

ele.l(nt)       = 0;
pn.l(n)         = 0;
pcarb.l         = 0;
pkx.l(t)$((-xtelec("capital",t))$shortrun) =1;

ra.l =     (-sam("capital","roi")+sum(xt,(-xtelec("capital",xt)))$(not shortrun))*pk.l
                + sum(xt$shortrun, (-xtelec("capital",xt))*pkx.l(xt))
                + sam("labor","ra")*pl.l
                + sum(ff,(-sam("rent",ff))*pr.l(ff))
                + sum(n, nrsupply(n)*pn.l(n))
                + (carblim*pcarb.l)$carblim
                + sum(xt$limit(xt), limit(xt)*plim.l(xt))
                - sum(t$quota(t), pele.l*ele.l(t)*tau.l)
                - (pc.l*phi_ls.l)$dd;
```

```
govt.l$dd =    (carblim*pcarb.l)$carblim + pc.l*phi_ls.l
                + (-sam("roi","ra")/(1+tc0$dd))*c.l*(pc.l/(proi.l*(1+tc0*phi_tc.l)$dd)
                  /(1+tc0$dd))**esub_c*pc.l*tc0*phi_tc.l;

*       Check the benchmark:
*       - marginal of all active activities must be zero
*       - marginal of all positivie prices must be zero
*       - marginal of all positive incomens must be zero

mcp_hybrid.iterlim = 0;
solve mcp_hybrid using mcp;

*       Relax iteration limit for counterfactual policy analysis
mcp_hybrid.iterlim      = 4000;

*===========================================================================
*       Analysis of policy scenarios (as laid out in the paper)
*
*       (i)   gradual nuclear phase-out
*       (ii)  target quota for renewables (green quota)
*       (iii) carbon taxation (environmental tax reform)


*       Define report parameters
parameter
        ev(*)           Equivalent variation in income
        supply(*,*)     Electricity supply by technology
        carbtax(*)      Carbon permit price
        subsidy         Subsidy rate on electricity from renewables
        report          Report default parameter;

scalar epsilon /1.e-5/;

*===========================================================================
*       Scenario 1: Gradual nuclear phase-out

set     nsc  Nuclear phase scenarios / 0, 25, 50, 75, 100/;

parameter limit_0       Base year capacity limits;
limit_0("nuclear") = limit("nuclear");

loop(nsc,
```

```

*       Assign available capacity for nuclear power
        limit("nuclear") = (1 - (ord(nsc)-1)/(card(nsc)-1))*limit_0("nuclear");
Display limit;
*       If nuclear capacity is set to zero, assure complete nuclear phase out
        if ((not limit("nuclear")),
           ele.fx("nuclear") = 0;
           );
        solve mcp_hybrid using mcp;
        supply(nsc,t)       = ele.l(t)  + epsilon;
        ev(nsc)             = 100 * (c.l-1) + epsilon ;

);


$setglobal labels nsc
$setglobal gp_opt0 "set data style linespoints"

$setglobal gp_opt1 "set key below"
report(nsc,"ev") = ev(nsc);
$setglobal gp_opt2 "set title 'Welfare changes'"
$setglobal gp_opt3 "set xlabel 'Nuclear capacity reduction (% vis--vis BaU)'"
$setglobal gp_opt4 "set ylabel 'Equivalent variation in income (%)'"
$libinclude plot report
display report;
report(nsc,"ev") = 0;

$setglobal gp_opt2 "set title 'Electricity supply by technology'"
$setglobal gp_opt3 "set xlabel 'Nuclear capacity reduction (% vis--vis BaU)'"
$setglobal gp_opt4 "set ylabel 'Activity level of technologies'"
$libinclude plot supply

*       Re-initialize parameterization for subsequent scenarios
limit("nuclear") = limit_0("nuclear");
ele.lo("nuclear") = 0; ele.up("nuclear") = +inf; ele.l("nuclear")=xtelec("ele","nuclear");


*========================================================================
*       Scenario 2: Green quotas

set     qsc  Green quota scenarios / 0 13, 5 18, 10 23, 15 28, 20 33/;
*       Note:   We start from the base year situation without binding target
*               share and then increase the share iteratively by 5%.
```

```
*               The descriptive text for scenario set elements captures
*               the actual target level of green electricity as percent
*               in overall electricity production (base year quota is 13%).
*               The plot-command picks up the descriptive text as
*               scenario labels when produce a graphical exposition of results.

*       Assign initial level values for variables
roi.l = 1; ele.l(xt)= xtelec("ele",xt);  s.l(ff) = 1; c.l = 1;
proi.l = 1; pele.l = 1; pf.l(ff) = 1; pl.l = 1; pk.l = 1; pr.l(ff) = 1;
pkx.l(t)$((-xtelec("capital",t))$shortrun) = 1; plim.l(t) = 0;
pn.l(n) = 1; pc.l = 1;
*       Install lower bounds on prices to avoid divison by zero in MCP formulation
proi.lo = 1e-5; pele.lo = 1e-5; pf.lo(ff) = 1e-5; pl.lo = 1e-5; pk.lo = 1e-5; pr.lo(ff) = 1e-5;
pkx.lo(t)$((-xtelec("capital",t))$shortrun) = 1e-5; pc.lo = 1e-5;
ra.l = c0;

parameter share_0       Base year renewable share;
share_0 = share;

quota(res) = yes;
tau.lo = 0; tau.l = 0; tau.up = 0.99;

loop(qsc,
*       Assign target shares for renewables in electricity production
        share  = min(1, (share_0 + 20/100* (ord(qsc)-1)/(card(qsc)-1)));

        solve mcp_hybrid using mcp;

        supply(qsc,t)   = ele.l(t)  + epsilon;
        ev(qsc)         = 100 * (c.l-1) + epsilon;
        subsidy(qsc)    = 100*tau.l + epsilon;
);

$setglobal labels qsc

report(qsc,"ev") = ev(qsc);
$setglobal gp_opt2 "set title 'Welfare changes'"
$setglobal gp_opt3 "set xlabel 'Green quota in % of overall electricity supply'"
$setglobal gp_opt4 "set ylabel 'Equivalent variation in income (%)'"
$libinclude plot report
display report;
report(qsc,"ev") = 0;

```

```
$setglobal gp_opt2 "set title ’Electricity supply by technology’"
$setglobal gp_opt3 "set xlabel ’Green quota in % of overall electricity supply’"
$setglobal gp_opt4 "set ylabel ’Activity level of technologies’"
$libinclude plot supply

report(qsc,"subsidy") = subsidy(qsc);
$setglobal gp_opt2 "set title ’Subsidy on renewables’"
$setglobal gp_opt3 "set xlabel ’Green quota in % of overall electricity supply’"
$setglobal gp_opt4 "set ylabel ’Subsidy rate (% of electricity price)’"
$libinclude plot report
display report;
report(qsc,"subsidy") = 0;

*       Re-initialize parameterization for subsequent scenarios
share = share_0;
quota(res) = no;
tau.fx = 0;
*===========================================================================
*       Scenario 3: Carbon taxation (double dividend)

*       First re-specify base year (benchmark) to public good extension
mcp_hybrid.iterlim      = 0;

dd  = 1;
g.lo = 0; g.up = + inf; govt.lo = 0; govt.up = + inf;
g0  = 0.2 *(-sam("roi","ra"));
tc0 = g0/((-sam("roi","ra")) - g0);
display g0, tc0;

*       Relax ficed variables
g.lo = 0; g.up = +inf;  pg.lo = 0; pg.up = +inf; govt.lo = 0; govt.up = + inf;
pcarb.lo = 0; pcarb.up = + inf;

*       Initially, we assume that lump-sum transfers are active
*       as the equal-yield instrument
phi_ls.l  = 0; phi_ls.lo = -inf; phi_ls.up = +inf;
phi_tc.fx = 1;

*       Assign base year carbon emissions (at shadow price of zero)
carblim = sum(ff, sam(ff,ff)*carbon(ff));
pcarb.l = 0;

*       Benchmark replication check for the model with public good extension
```

```
*       Initialize activities and prices
roi.l = 1; ele.l(xt)= xtelec("ele",xt);  ele.l(nt) = 0; s.l(ff) = 1; c.l = 1;
proi.l = 1; pele.l = 1; pf.l(ff) = 1; pl.l = 1; pk.l = 1; pr.l(ff) = 1; pg.l = 1;
pkx.l(t)$((-xtelec("capital",t))$shortrun) = 1; plim.l(t)$limit(t) = 0;
pc.l = 1; pn.l(n) = 0; ra.l = c0; govt.l = g0;

*       Install lower bounds on prices to avoid divison by zero in MCP formulation
proi.lo = 1e-5; pele.lo = 1e-5; pf.lo(ff) = 1e-5; pl.lo = 1e-5; pk.lo = 1e-5; pr.lo(ff) = 1e-5;
pkx.lo(t)$((-xtelec("capital",t))$shortrun) = 1e-5; pc.lo = 1e-5; pg.lo = 1e-5;

*       Check the re-specified benchmark:
*       - marginal of all active activities must be zero
*       - marginal of all positivie prices must be zero
*       - marginal of all positive incomens must be zero

mcp_hybrid.iterlim = 0;

solve mcp_hybrid using mcp;

*       Relax iteration limit
mcp_hybrid.iterlim = 4000;


*       Specification of carbon tax scenarios based on exogenous emission reduction targets
set     csc  Carbon abatement scenarios scenarios / 0, 5, 10, 15, 20/;

parameter carbon_0      Benchmark capacity limits;
parameter ev_           Report parameter for welfare changes;

carbon_0 = carblim;

display carbon_0;

loop(csc,
*       Assign carbon emission limit
        carblim = (1 - 0.2*(ord(csc)-1)/(card(csc)-1))*carbon_0;

*       Activate lump-sum transfer as recycling instrument
phi_ls.l  = 0; phi_ls.lo = -inf; phi_ls.up = +inf;
phi_tc.fx = 1;

        solve mcp_hybrid using mcp;

```

```
        ev_(csc,"ls")           = 100 * (c.l-1) + epsilon;

*       Activate consumption tax as recycling instrument
phi_tc.l = 1; phi_tc.lo = -0.99;  phi_tc.up = +inf;
phi_ls.fx  = 0;
        solve mcp_hybrid using mcp;

        ev_(csc,"tc")           = 100 * (c.l-1) + epsilon;
);

$setglobal labels csc
$setglobal gp_opt2 "set title 'Welfare changes'"
$setglobal gp_opt3 "set xlabel 'Carbon emission reduction (in % vis--vis base year)'"
$setglobal gp_opt4 "set ylabel 'Equivalent variation in income (%)'"
$libinclude plot ev_
display ev_;
ev_(csc,"tc") = 0; ev_(csc,"ls") = 0;

*       Re-initialize parameterization for subsequent policy scenarios
dd = 0; g0 = 0; tc0 = 0;
```

Part 2: MPSGE Formulation

```
$Title  Static maquette of integrated TD/BU hybrid model

*       Model formulation in meta-language MPSGE
*       (see Rutherford 1995 for documentation)

*=============================================================================
* Model code for stylzed integrated bottom-up/top-down analysis of energy
* policies based on:
*
*               ZEW Discussion Paper 05-28
*          Integrating Bottom-Up into Top-Down:
*            A Mixed Complementarity Approach
*
* Contact the authors at: boehringer@zew.de; tom@mpsge.org
*==============================================================================

* For plotting the results you must have installed the gnuplot-shareware
* (see http://debreu.colorado.edu/gnuplot/gnuplot.htm for downloads)

*==============================================================================
*       List of parameters subject to sensitivity analysis
*       The user can change the default settings.

* Choice of key elasticities:
*       Elasticity of substitution in final consumption
$if not setglobal esub_c $setglobal esub_c 0.5

*       Elasticity in gas supply
$if not setglobal esub_gas  $setglobal esub_gas  1.5

*       Elasticity in coal supply
$if not setglobal esub_coal  $setglobal esub_coal 3

*       Elasticity in oil supply
$if not setglobal esub_oil  $setglobal esub_oil 1.5


* Choice of resource availability for renewables:
* (as a fraction of base-year total electricity production)
*       Potential wind supply - (%)
$if not setglobal p_wind  $setglobal p_wind  10
```

```
*       Potential solar supply - (%)
$if not setglobal p_sun  $setglobal p_sun   10

*       Potential biomass supply - (%)
$if not setglobal p_trees  $setglobal p_trees 10


* Cost disadvantage of inital slack technologies:
*       Wind energy premium (%)
$if not setglobal c_wind  $setglobal c_wind 10

*       Solar energy premium (%)
$if not setglobal c_solar  $setglobal c_solar   10

*       Biomass energy premium (%)
$if not setglobal c_biomass  $setglobal c_biomass 10


* Other central model assumptions:
*       Time horizon (short, long)
*       N.B.: For short-run analysis capital is immobile across sectors
$if not setglobal horizon  $setglobal horizon long

*=============================================================================


*       Assign user-specific changes of default assumptions
scalar  shortrun        Flag for short-run capital mobility/1/;

$if "%horizon%"=="long" shortrun=0;

*       Elasticitities of substitution (ESUB)
scalar  esub_c     Elasticity of substituion in final demand /%esub_c%/
        esub_ele   ESUB between electricity and oil in final demand /0.5/
        esub_k_e   ESUB between capital and energy in ROI production /0.5/
        esub_l_ke  ESUB between labor and other inputs in ROI production /0.8/;

set     t       Electricity Technologies (current and future)
                /coal,gas,nuclear,hydro,wind,solar,biomass/;

set     xt(t)   Existing technologies /coal,gas,nuclear,hydro/;
```

```
set     nt(t)   New vintage technologies /wind,solar,biomass/;

set     ff      Fossil fuel inputs /coa, gas, oil/;

set     n       Natural resources /wind, sun, trees/;


set     res(t)  Renewable energy sources /hydro, wind, solar, biomass/;

*       The following data table describes an economic equilibrium in
*       the base year:


table sam  Base year social accounting matrix

        roi     coa     gas     oil     ele     ra
roi     200     -5      -5      -10     -10    -170
coa             15                      -15
gas                     15              -15
oil                             30              -30
ele     -10                              60     -50
capital -80                             -20     100
labor   -110    -5      -5      -10             130
rent            -5      -5      -10              20             ;

parameter carbon(ff)    Carbon coefficients /oil 1, gas 1, coa 2/;

scalar  carblim         Carbon target /0/;

parameter  esub_ff(ff)  Elastictity of substitution in fossil fuel production
        /gas %esub_gas%, coa %esub_coal%, oil %esub_oil%/;

*       The following data tables describes electricy generation in
*       the base year as well as the technology coefficients for technologies
*       which are inactive in the base year (wind, solar, biomass). Inactive
*       technologies are by defaults %c_****% more costly.

table xtelec  Electricty technologies - extant (initially active)

        coal    gas     nuclear hydro
ele       20     20          12     8
roi       -1     -1          -8
gas             -15
```

```
coa      -15
capital   -4     -4           -4    -8;


table ntelec  Electricty technologies - new vintage (initially inactive)

         wind    solar   biomass
ele       1.0      1.0      1.0
roi       -.2      -.3      -.4
capital   -.9      -.8      -.7
wind     -1.0
sun               -1.0
trees                      -1.0;


*       Adjust the cost coefficients for initially inactive technologies
*       according to user assumptions:
set xk /roi, capital/;

ntelec(xk,"wind")    = ntelec(xk,"wind")    * (100+%c_wind%)/110;
ntelec(xk,"solar")   = ntelec(xk,"solar")   * (100+%c_solar%)/110;
ntelec(xk,"biomass") = ntelec(xk,"biomass") * (100+%c_biomass%)/110;


*       Specify limits (resource or policy constraints) to the availability
*       of technologies

parameter limit    Electricty supply limits on extant technologies /
        nuclear         12
        hydro            8 /;

parameter  nrsupply(n)  Natural resource supplies (fraction of base output)/
        wind            %p_wind%
        sun             %p_sun%
        trees           %p_trees% /;

nrsupply(n) = nrsupply(n)/100 * sam("ele","ele");

parameter c0    Baseyear final consumption;
c0 = (-sam("roi","ra")-sam("ele","ra")-sam("oil","ra"));


set     quota(t)  Flag for technologies contributing to green quota;
```

```
quota(t) = no;

scalar  share   Target share for green quota /0/;

*       By default we might set target share for green quota at base year level
share = sum(t$res(t), xtelec("ele",t))/sum(t, xtelec("ele",t));
display share;

scalar
        dd      Flag for double dividend policy analysis /0/,
        ls      Flag for lump-sum revenue-recyling       /0/,
        vat     Flag for VAT revenue recycling           /0/,
        g0      Base year public consumption             /0/,
        tc0     Base year consumption tax                /0/;

*       MPSGE formulation of the hybrid model
$ontext

$model:mps_hybrid

$sectors:
        roi     ! Aggregate output
        ele(t)  ! Production levels for electricity by technology
        s(ff)   ! Fossil fuel supplies
        c       ! Aggregate consumption (utility) formation
        g$dd    ! Public good provision

$commodities:
        proi    ! Price of aggregate output
        pele    ! Price of electricty
        pf(ff)  ! Price of oil and gas
        pl      ! Wage rate
        pk      ! Price of malleable capital for X (and NT elec)
        pr(ff)  ! Rent on fossil fuel resources
        pn(n)   ! Rent on natural resources
        pc      ! Consumption (utility) price index
        pg$dd   ! Price of public consumption
        plim(t)$limit(t)                        ! Shadow price on electricity expansion
        pkx(t)$((-xtelec("capital",t))$shortrun) ! Price of capital to extant technologies
        pcarb$carblim                           ! Carbon tax rate

$consumers:
        ra      ! Representative household
```

```
        govt$dd ! Government

$auxiliary:
        tau$card(quota)  ! Uniform subsidy rate on renewable energy
        phi_ls$dd        ! Lump-sum recycling
        phi_tc$dd        ! Consumption tax recycling


*       Aggregate output:

$prod:roi s:esub_l_ke ke:esub_k_e
        o:proi  q:sam("roi","roi")
        i:pl    q:(-sam("labor","roi"))
        i:pk    q:(-sam("capital","roi"))  ke:
        i:pele  q:(-sam("ele","roi"))      ke:

*       Extant electricity:

$prod:ele(xt)
        o:pele              q:1   raa:ra   n:tau$quota(xt)     m:(-1)$quota(xt)
        i:proi              q:(-xtelec("roi",xt)/xtelec("ele",xt))
        i:pf(ff)            q:(-xtelec(ff,xt)/xtelec("ele",xt))
        i:pkx(xt)$shortrun  q:(-xtelec("capital",xt)/xtelec("ele",xt))
        i:pk$(not shortrun) q:(-xtelec("capital",xt)/xtelec("ele",xt))
        i:plim(xt)$limit(xt) q:1

*       New vintage electricity:

$prod:ele(nt)
        o:pele  q:1     a:ra   n:tau$quota(nt)  m:(-1)$quota(nt)
        i:proi  q:(-ntelec("roi",nt))
        i:pk    q:(-ntelec("capital",nt))
        i:pn(n) q:(-ntelec(n,nt))

$prod:s(ff) s:0  r:esub_ff(ff)   xl(r):0
        o:pf(ff)        q:sam(ff,ff)
        i:pcarb$carblim q:(carbon(ff)*sam(ff,ff))
        i:proi          q:(-sam("roi",ff))      xl:
        i:pl            q:(-sam("labor",ff))    xl:
        i:pr(ff)        q:(-sam("rent",ff))     r:

$prod:c s:esub_c e:esub_ele
        o:pc            q:c0
```

```
        i:proi          q:(-sam("roi","ra")/(1+tc0$dd)) p:(1+tc0$dd)
+ a:govt$dd n:phi_tc$dd  m:tc0$dd
        i:pele          q:(-sam("ele","ra"))    e:
        i:pf("oil")     q:(-sam("oil","ra"))    e:

$demand:ra
        d:pc                      q:c0
        e:pk    q:(-sam("capital","roi")+sum(xt,(-xtelec("capital",xt)))$(not shortrun))
        e:pkx(xt)$shortrun        q:(-xtelec("capital",xt))
        e:pl                      q:(sam("labor","ra"))
        e:pr(ff)                  q:(-sam("rent",ff))
        e:pn(n)                   q:nrsupply(n)
        e:pcarb$carblim$(not dd)  q:carblim
        e:plim(xt)$limit(xt)      q:limit(xt)
        e:pc$dd                   q:(-1)        r:phi_ls$dd


$demand:govt$dd
        d:pg    q:g0
        e:pc    q:1             r:phi_ls
        e:pcarb$carblim         q:carblim

$prod:g$dd
        O:pg    q:g0
        i:proi  q:g0


$constraint:tau$card(quota)
        sum(t$res(t), ele(t)) =e= share*sum(t, ele(t));


$constraint:phi_ls$dd
        g =e= 1;

$constraint:phi_tc$dd
        g=e= 1;

$offtext
$sysinclude mpsgeset mps_hybrid

*       In the base year new-vintage technologies are inactive
*       and the prices of backstop natural resources are zero
*       Extant technologies with capacity limits are assumed to
```

```
*       operate at the upper bound with a zero shadow value in the
*       base year

ele.l(nt)  = 0;
pn.l(n)    = 0;
plim.l(xt) = 0;

ele.l(xt)  = xtelec("ele",xt);

*       Benchmark replication check
mps_hybrid.iterlim = 0;
$include mps_hybrid.gen
solve mps_hybrid using mcp;

display "The precision of the benchmark dataset is:", mps_hybrid.objval;
abort$(ABS(mps_hybrid.objval) gt 1e-4)"MPSGE model does not calibrate";


*       Relax iteration limit for counterfactual policy analysis
mps_hybrid.iterlim      = 4000;


*============================================================================
*       Analysis of policy scenarios (as laid out in the paper)
*
*       (i)   gradual nuclear phase-out
*       (ii)  target quota for renewables (green quota)
*       (iii) carbon taxation (environmental tax reform)


*       Define report parameters
parameter
        ev(*)           Equivalent variation in income
        supply(*,*)     Electricity supply by technology
        carbtax(*)      Carbon permit price
        subsidy         Subsidy rate on electricity from renewables
        report          Report default parameter;

scalar epsilon /1.e-5/;


*============================================================================
*       Scenario 1: Gradual nuclear phase-out
```

```
set     nsc  Nuclear phase scenarios / 0, 25, 50, 75, 100/;

parameter limit_0       Base year capacity limits;
limit_0("nuclear") = limit("nuclear");

loop(nsc,

*       Assign available capacity for nuclear power
        limit("nuclear") = (1 - (ord(nsc)-1)/(card(nsc)-1))*limit_0("nuclear");
*       If nuclear capacity is set to zero, assure complete nuclear phase out
        if ((not limit("nuclear")),
           ele.fx("nuclear") = 0;
           );
$include mps_hybrid.gen
        solve mps_hybrid using mcp;
        supply(nsc,t)       = ele.l(t)  + epsilon;
        ev(nsc)             = 100 * (c.l-1) + epsilon ;

);


$setglobal labels nsc
$setglobal gp_opt0 "set data style linespoints"

$setglobal gp_opt1 "set key below"
report(nsc,"ev") = ev(nsc);
$setglobal gp_opt2 "set title ’Welfare changes’"
$setglobal gp_opt3 "set xlabel ’Nuclear capacity reduction (% vis--vis BaU)’"
$setglobal gp_opt4 "set ylabel ’Equivalent variation in income (%)’"
$libinclude plot report
display report;
report(nsc,"ev") = 0;


$setglobal gp_opt2 "set title ’Electricity supply by technology’"
$setglobal gp_opt3 "set xlabel ’Nuclear capacity reduction (% vis--vis BaU)’"
$setglobal gp_opt4 "set ylabel ’Activity level of technologies’"
$libinclude plot supply

*       Re-initialize parameterization for subsequent scenarios
limit("nuclear") = limit_0("nuclear");
ele.lo("nuclear") = 0; ele.up("nuclear") = +inf;
```

```
ele.l("nuclear")=xtelec("ele","nuclear");


*============================================================================
*       Scenario 2: Green quotas

set     qsc  Green quota scenarios / 0 13, 5 18, 10 23, 15 28, 20 33/;
*       Note:   We start from the base year situation without binding target
*               share and then increase the share iteratively by 5%.
*               The descriptive text for scenario set elements captures
*               the actual target level of green electricity as percent
*               in overall electricity production (base year quota is 13%).
*               The plot-command picks up the descriptive text as
*               scenario labels when produce a graphical exposition of results.


parameter share_0       Base year renewable share;
share_0 = share;

quota(res) = yes;


loop(qsc,
*       Assign target shares for renewables in electricity production
        share  = min(1, (share_0 + 20/100* (ord(qsc)-1)/(card(qsc)-1)));

$include mps_hybrid.gen
        solve mps_hybrid using mcp;

        supply(qsc,t)   = ele.l(t)  + epsilon;
        ev(qsc)         = 100 * (c.l-1) + epsilon;
        subsidy(qsc)    = 100*tau.l + epsilon;
);

$setglobal labels qsc

report(qsc,"ev") = ev(qsc);
$setglobal gp_opt2 "set title 'Welfare changes'"
$setglobal gp_opt3 "set xlabel 'Green quota in % of overall electricity supply'"
$setglobal gp_opt4 "set ylabel 'Equivalent variation in income (%)'"
$libinclude plot report
display report;
report(qsc,"ev") = 0;
```

```
$setglobal gp_opt2 "set title 'Electricity supply by technology'"
$setglobal gp_opt3 "set xlabel 'Green quota in % of overall electricity supply'"
$setglobal gp_opt4 "set ylabel 'Activity level of technologies'"
$libinclude plot supply

report(qsc,"subsidy") = subsidy(qsc);
$setglobal gp_opt2 "set title 'Subsidy on renewables'"
$setglobal gp_opt3 "set xlabel 'Green quota in % of overall electricity supply'"
$setglobal gp_opt4 "set ylabel 'Subsidy rate (% of electricity price)'"
$libinclude plot report
display report;
report(qsc,"subsidy") = 0;

*       Re-initialize parameterization for subsequent scenarios
share = share_0;
quota(res) = no;
$exit
*===========================================================================
*       Scenario 3: Carbon taxation (double dividend)

*       First re-calibrate base year (benchmark) to public good extension
mps_hybrid.iterlim      = 0;

dd  = 1;
g0  = 0.2 *(-sam("roi","ra"));
tc0 = g0/((-sam("roi","ra")) - g0);
display g0, tc0;

*       Initially, we assume that lump-sum transfers are active
*       as the equal-yield instrument
phi_ls.l  = 0; phi_ls.lo = -inf; phi_ls.up = +inf;
phi_tc.fx = 1;

*       Assign base year carbon emissions (at shadow price of zero)
carblim = sum(ff, sam(ff,ff)*carbon(ff));
pcarb.l = 0;

*       Benchmark replication check for the model with public good extension
*       Initialize activities and prices
roi.l = 1; ele.l(xt)= xtelec("ele",xt);  ele.l(nt) = 0; s.l(ff) = 1; c.l = 1;
proi.l = 1; pele.l = 1; pf.l(ff) = 1; pl.l = 1; pk.l = 1; pr.l(ff) = 1;
pkx.l(t)$((-xtelec("capital",t))$shortrun) = 1; plim.l(t)$limit(t) = 0;
```

```
pc.l = 1; pn.l(n) = 0; ra.l = c0; govt.l = g0;

mps_hybrid.iterlim = 0;
$include mps_hybrid.gen
solve mps_hybrid using mcp;

display "The precision of the re-specified benchmark dataset is:", mps_hybrid.objval;
abort$(ABS(mps_hybrid.objval) gt 1e-4)"MPSGE model does not calibrate";

*       Relax iteration limit
mps_hybrid.iterlim = 4000;


*       Specification of carbon tax scenarios based on exogenous emission reduction targets
set     csc  Carbon abatement scenarios scenarios / 0, 5, 10, 15, 20/;

parameter carbon_0      Benchmark capacity limits;
parameter ev_           Report parameter for welfare changes;

carbon_0 = carblim;

display carbon_0;

loop(csc,
*       Assign carbon emission limit
        carblim = (1 - 0.2*(ord(csc)-1)/(card(csc)-1))*carbon_0;

*       Activate lump-sum transfer as recycling instrument
phi_ls.l  = 0; phi_ls.lo = -inf; phi_ls.up = +inf;
phi_tc.fx = 1;
$include mps_hybrid.gen
        solve mps_hybrid using mcp;
        ev_(csc,"ls")           = 100 * (c.l-1) + epsilon;

*       Activate consumption tax as recycling instrument
phi_tc.l = 1; phi_tc.lo = -0.99;  phi_tc.up = +inf;
phi_ls.fx  = 0;
$include mps_hybrid.gen
        solve mps_hybrid using mcp;
        ev_(csc,"tc")           = 100 * (c.l-1) + epsilon;
);

$setglobal labels csc
```

```
$setglobal gp_opt2 "set title 'Welfare changes'"
$setglobal gp_opt3 "set xlabel 'Carbon emission reduction (in % vis--vis base year)'"
$setglobal gp_opt4 "set ylabel 'Equivalent variation in income (%)'"
$libinclude plot ev_
display ev_;
ev_(csc,"tc") = 0; ev_(csc,"ls") = 0;

*       Re-initialize parameterization for subsequent policy scenarios
dd = 0; g0 = 0; tc0 = 0;
```